
Lineárńı algebra

1. Možnosti baĺıku linalg
Před použit́ım jakéhokoliv př́ıkazu lineárńı algebry je nutno vyvolat speciálńı baĺık programů linalg
určený pro lineárńı algebru.

> restart;

> with(linalg):

Ukonč́ıte-li př́ıkaz středńıkem, Maple vyṕı̌se včechny př́ıkazy, které jsou součást́ı baĺıku linalg. Ukonč́ıte-
li př́ıkaz dvojtečkou, Maple vás upozorńı:

Warning, the protected names norm and trace have been redefined and unprotected

(př́ıkaz norm, je-li vyvolán v rámci baĺıku linalg, poč́ıtá normu vektoru nebo matice, př́ıkaz trace poč́ıtá
v rámci baĺıku linalg stopu matice).

1.1. Práce s vektory a maticemi

Uved’me si nejprve několik př́ıklad̊u, pomoćı kterých vytvoř́ıme vektor a matici. Povšimněte si, že
př́ıkaz transpose vytvoř́ı k dané matici matici transponovanou, ale vektory ponechá v řádkovém tvaru.
Přesto, pokud budete matici násobit vektorem, muśı vždy rozměry (zjist́ı je př́ıkazy rowdim a coldim)
odpov́ıdat, t.j. násob́ım-li matici A matićı B zprava, muśı mı́t matice A stejný počet sloupc̊u jako matice
B řádk̊u (a toto plat́ı i v př́ıpadě, že B je vektor), přičemž výsledná matice má počet řádk̊u stejný jako
matice A a počet sloupc̊u stejný jako matice B:

> u := [3,2,3,5];

u := [3, 2, 3, 5]

> v := vector([3,2,1,9]);

v := [3, 2, 1, 9]

> v1:= vector[row]([3,2,1,9]);

v1 := [3, 2, 1, 9]

> v2:=array([1,-1,2]);

v2 := [1, −1, 2]

> w := transpose(v2);

w := transpose(v2 )

> C := matrix( [[1,2,3]] );

C :=
[

1 2 3
]

> Ct := transpose(C);

Ct :=

 1
2
3


> rowdim(Ct);

3

> coldim(Ct);

1

> U := array(diagonal, 1..2,1..3, [(1,1)=5]);

U :=

[
5 0 0
0 U2, 2 0

]

Předchoźı př́ıklad vytvořil diagonálńı obdélńıkovou matici U typu 2×3, přičemž jsme č́ıselnou hodnotu
zadali pouze pro prvek U1,1.

> multiply(U,C);

Error, (in multiply) non matching dimensions for vector/matrix
product

> multiply(U,Ct); [
5

2U2, 2

]
Složky vektoru lze také definovat jako funkce nezávisle proměnné:

> f:= (j) -> x^(j-1):
> vector(5,f);

[1, x, x2, x3, x4]

Matici vytvoř́ıme např́ıklad tak, že ji zadáme po řádćıch výčtem jej́ıch prvk̊u. Př́ıkaz submatrix
vybere př́ıslušnou podmatici, př́ıkazy delrows a delcols oř́ıznou uvedené řádky a sloupce, př́ıkaz
extend matici naopak rozš́ı̌ŕı o uvedený počet řádk̊u a sloupc̊u. Povšimněte si, že do všech nově
vzniklých řádk̊u a sloupc̊u dosad́ı př́ıkaz extend stejná č́ısla (posledńı parametr). Chceme-li, aby
nově vzniklé prvky byly zadány jako vektory, muśıme použ́ıt př́ıkazy augment (spoj́ı dvě matice ”za
sebou”) a stackmatrix (spoj́ı dvě matice ”pod sebou”) . Pozor na rozměry!

> A := matrix([[0,5,4,5],[5,-1,4,4],[6,5,-3,7]]);

A :=

 0 5 4 5
5 −1 4 4
6 5 −3 7


> A[2,3];

4

> B:=submatrix(A, 1..2, 2..4);

B :=

[
5 4 5
−1 4 4

]
> E:=delrows(A, 2..3);

E :=
[

0 5 4 5
]

> E:=delcols(E,2..4);

E :=
[

0
]

> E:=extend(E, 1, 1, 1);

E :=

[
0 1
1 1

]
> Bt:=transpose(B);

Bt :=

 5 −1
4 4
5 4


> augment(A,Bt);  0 5 4 5 5 −1

5 −1 4 4 4 4
6 5 −3 7 5 4


> B1:=extend(B,0,1,0);

B1 :=

[
5 4 5 0
−1 4 4 0

]
> stackmatrix(A,B1); 

0 5 4 5
5 −1 4 4
6 5 −3 7
5 4 5 0
−1 4 4 0


Pokusme se změnit některé prvky už existuj́ıćı matice A, např́ıklad chceme v matici A nahradit v
prvńım a druhém řádku třet́ı a čtvrtý sloupec matićı E:

> copyinto(E,A,1,3);  0 5 0 1
5 −1 1 1
6 5 −3 7


> A;

A

Baĺık linalg zacháźı s vektory a maticemi jako s funkcemi, nikoliv jako s proměnnými, které maj́ı
hodnotu. Vektor nebo matici muśıte ”vyč́ıslit” pomoćı př́ıkazu evalm:

> evalm(A);  0 5 0 1
5 −1 1 1
6 5 −3 7



1.2. Základńı vektorové a maticové operace

Definujme dva vektory u a v a dvě matice A a B a ukažme si na nich, jak vypadá v Maplu (baĺık
linalg) sč́ıtáńı a násobeńı těchto objekt̊u.

> restart;

> with(linalg):

> u:=vector([-1,2,5]);

u := [−1, 2, 5]

> v:= vector([2,1,-2]);

v := [2, 1, −2]

> u+v;

u+ v

> evalm(%);

[1, 3, 3]

> matadd(u,v);

[1, 3, 3]

Vid́ıme, že sč́ıtáme-li dva vektory, znaménko + nevyč́ısĺı výsledný vektor a chceme-li ho znát, muśıme
ještě použ́ıt funkci evalm. Použijeme-li př́ıkaz matadd, dostaneme rovnou vyč́ıslený výsledný vektor.
Obdobně pro matice:

> A:=matrix(2,3,[5,4,1,6,-1,3]);

A :=

[
5 4 1
6 −1 3

]
> B:=matrix(2,3,[1,-1,4,2,3,1]);

B :=

[
1 −1 4
2 3 1

]
> A+B;

A+B

> evalm(%); [
6 3 5
8 2 4

]
> matadd(A,B); [

6 3 5
8 2 4

]
Jak je to s násobeńım dvou vektor̊u?

> u*v;

u v

> evalm(%);

Error, (in evalm/evaluate) use the &* operator for matrix/vector
multiplication

Jak rad́ı nápověda chybového hlášeńı, neńı operátor pro maticové násobeńı jenom ∗, ale je &∗.
Použijeme ho, ale č́ıselnou hodnotu př́ıslušného součinu dostaneme teprve, vyvoláme-li funkci evalm.

Pozor, pokud násob́ım takto dva vektory, výsledkem je jejich skalárńı součin. Násob́ıme-li dvě matice,
muśıme dbát na jejich rozměry.

> u&*v;

u&∗ v
> evalm(%);

−10

> A&*(transpose(B));

A&∗

 1 2
−1 3

4 1


> evalm(%); [

5 23
19 12

]
Baĺık linalg umožňuje ještě daľśı vektorová a maticová násobeńı. Př́ıkaz dotprod poč́ıtá skalárńı
součin dvou vektor̊u, př́ıkaz multiply vypočte součin matic, přičemž je třeba dbát, aby obě násobené
matice měly správné rozměry. Aplikujeme-li multiply na dva vektory, př́ıkaz vraćı skalárńı součin.
Př́ıkaz crossprod vypočte vektorový součin dvou vektor̊u z R3, scalarmul násob́ı matici nebo vek-
tor algebraickým výrazem, innerprod násob́ı posloupnost vektor̊u a matic; opět je třeba dbát aby
násobené matice měly správné rozměry. Pokud posloupnost sestává pouze ze dvou vektor̊u, vypočte
se jejich skalárńı součin.

> dotprod(u,v);

−10

> multiply(u,v);

−10

> multiply(transpose(A),B);  17 13 26
2 −7 15
7 8 7


> innerprod(u,v);

−10

> innerprod(u,transpose(A),B);

[22, 13, 39]

> innerprod(u,transpose(A),B,v);

−21

> crossprod(u,v);

[−9, 8, −5]

> scalarmul(v, x);

[2x, x, −2x]

> scalarmul(A,2*x); [
10x 8x 2x
12x −2x 6x

]
Baĺık (linalg) umı́ také spoč́ıtat délku (normu) vektoru, přǐradit vektoru př́ıslušný vektor délky 1
(normalizovat daný vektor) a určit úhel, který sv́ıraj́ı dva vektory

> w:=vector[row]([1,2,2,4]);

w := [1, 2, 2, 4]

> nw:=([norm(w), norm(w,1), norm(w, 2),
> norm(w, infinity), norm(w,frobenius)]);

nw := [4, 9, 5, 4, 5]

Jaké ”délky” vektoru w = (w1, w2, . . . , wn)T jsme dostali? Př́ıslušný př́ıkaz baĺıku linalg má tvar
norm(vector,normname) nebo norm(matrix,normname). Parametr
normname má následuj́ıćı význam:

1. pro vektory je normname = a, kde a je přirozené č́ıslo, a ≥ 1, nebo a =∞ nebo a = frobenius :
||w||k = (|w1|k + |w2|k + . . .+ |wn|k)1/k;
||w||∞ = max(|w1|, |w2|, . . . , |wn|);
||w||F =

√
|w1|2 + |w2|2 + . . .+ |wn|2;

pokud nezadáme parametr normname, vyč́ısluje se ||w||∞;

2. pro matice muśı být normname 1, 2,∞ nebo frobenius, přičemž pro A = (aij), i = 1, . . . ,m; j =
1, . . . , n je
||A||1 = maxj (

∑m
i=1 |aij|) (= maximálńı sloupcový součet absolutńıch hodnot);

||A||2 je odmocnina maximálńıho vlastńıho č́ısla matice AAT;

||A||∞ = maxi

(∑n
j=1 |aij|

)
(= maximálńı řádkový součet absolutńıch hodnot);

||A||F =

√∑
i,j

|aij|2;

nezadáme-li parametr normname, vyč́ısĺı se stejně jako u vektor̊u ||A||∞.

Vypočtěme normu následuj́ıćı, tzv. Hilbertovy matice:

> Hilb:= (i,j) -> 1/(i+j-1):
> H:=matrix(4,4,Hilb);

H :=



1
1

2

1

3

1

4
1

2

1

3

1

4

1

5
1

3

1

4

1

5

1

6
1

4

1

5

1

6

1

7


> nH:=([norm(H), norm(H,1), norm(H, 2),
> norm(H, infinity), norm(H,
> frobenius)]);

nH := [
25

12
,

25

12
,
1

4
RootOf(

1− 6684916 Z + 9214816950 Z 2 − 20354692500 Z 3 + 558140625 Z 4,

index = 4)(1/2),
25

12
,

√
100517

210
]

> evalf(%);

[2.083333333, 2.083333333, 1.500214280, 2.083333333, 1.509734100]

> normalize(row(H,3));

[√
869
√

3600

2607
,

√
869
√

3600

3476
,

√
869
√

3600

4345
,

√
869
√

3600

5214

]
> normalize(col(H,3));[√

869
√

3600

2607
,

√
869
√

3600

3476
,

√
869
√

3600

4345
,

√
869
√

3600

5214

]
> angle(row(H,2),col(H,3));

arccos(
1200

√
1669

√
869

1450361
)

> evalf(%);

0.08462159349

> angle( vector([1,0]), vector([0,1]) );

π

2
Př́ıkaz angle poč́ıtá (v radiánech) úhel α dvou n−dimenzionálńıch vektor̊u u a v podle vzorce cos α =
u · v
|u||v|

.

Baĺık linalg umožňuje také pomoćı př́ıkazu basis vybrat bázi vektorového prostoru ze seznamu
vektor̊u nebo naj́ıt bázi vektorového prostoru, který generuj́ı řádky nebo sloupce dané matice.

> restart;

> with(linalg):

> v1 := vector( [1,0,1] ): v2 := vector(
> [0,-1,0] ): v3 := vector( [1,0,-1] ): v4 := vector( [1,1,1] ):
> basis( {v1, v2, v3} );

{v1 , v2 , v3}
> A := matrix([[0,5,4,5],[5,-1,4,4],[6,5,-3,7]]);

A :=

 0 5 4 5
5 −1 4 4
6 5 −3 7


> basis(A,’rowspace’);

[[0, 5, 4, 5], [5, −1, 4, 4], [6, 5, −3, 7]]

> basis(A,’colspace’);

[[0, 5, 6], [5, −1, 5], [4, 4, −3]]

Př́ıkaz nullspace vrát́ı bázi nulového prostoru dané matice, př́ıpadně jeho dimenzi. Připomeňme, že
nulový prostor N matice A typu m× n je lineárńı prostor,

N = {x ∈ Rn, Ax = 0}.

V následuj́ıćım př́ıkladu se do parametru d (volitelný parametr) dosad́ı dimenze nulového prostoru
matice A.

> nullspace(A,d);

{
[
1,

239

223
,

100

223
,
−319

223

]
}

> d;

1

Hodnost dané matice poč́ıtáme pomoćı př́ıkazu rank. Pozor, neplet’te si tento př́ıkaz s př́ıkazy rowdim
a coldim, které dávaj́ı jen počet řádk̊u a sloupc̊u dané matice, nehodnot́ı jejich lineárńı závislost a
nezávislost.

> rank(A);

3

1.3. Řešeńı soustav lineárńıch algebraických rovnic

Řešme nejprve soustavu dvou rovnic o dvou neznámých:

2x+ y = 0, x+ y = 1.

V tomto jednoduchém př́ıpadě lze hledané řešeńı snadno graficky znázornit, nebot’ hledáme společné
body př́ımek y = −2x a y = −x+ 1.

> restart;

> with(plots):with(linalg):

> implicitplot({2*x+y=0, x+y=1},x=-2..2,y=-1..3);

–1

1

2

3

y

–2 –1 1 2
x

V Maplu slouž́ı k řešeńı soustavy lineárńıch algebraických rovnic př́ıkaz solve:

> solve({2*x+y=0, x+y=1},{x,y});
{x = −1, y = 2}

Toto byl nejjednodušš́ı př́ıpad, kdy soustava měla právě jedno řešeńı. Mohou ale nastat ještě dva jiné
př́ıpady: soustava nemá řešeńı nebo soustava má nekonečně mnoho řešeńı:

> solve({x+y=2, x+y=1},{x,y});
> implicitplot({x+y=2, x+y=1},x=-2..2,y=-1..3);

–1

1

2

3

y

–2 –1 1 2
x

V tomto př́ıpadě soustava nemá řešeńı, nebot’ hledáme vlastně pr̊useč́ık dvou rovnoběžných př́ımek.
Př́ıkaz solve nevrátil žádnou hodnotu x a y.

> solve({2*x+2*y=2, x+y=1},{x,y});
{y = y, x = −y + 1}

> implicitplot({2*x+2*y=2, x+y=1},x=-2..2,y=-1..3);

–1

1

2

3

y

–2 –1 1 2
x

Tato soustava má nekonečně mnoho řešeńı, nebot’ obě rovnice jsou rovnicemi téže př́ımky.

Řešeńı soustavy tř́ı rovnic o třech neznámých představuje geometricky hledáńı společných bod̊u tř́ı
rovin. I zde soustava může mı́t jedno řešeńı, nekonečně mnoho řešeńı nebo žádné řešeńı.

> solve({x+y+z=0,y+z=1,x+y-z=1},{x,y,z});

{z =
−1

2
, y =

3

2
, x = −1}

> implicitplot3d({x+y+z=0,x+z=1,x+y-z=1},x=-3..3,y=-3..3,z=-3..3);

Všechny tři roviny maj́ı jeden společný bod. V následuj́ıćım př́ıkladu je pr̊usečnićı tř́ı rovin př́ımka,
soustava má tedy nekonečně mnoho řešeńı.

> solve({x+y+z=1,x-y+2*z=1,2*x+3*z=2},{x,y,z});
{z = 2 y, y = y, x = 1− 3 y}

> implicitplot3d({x+y+z=1,x-y+2*z=1,2*x+3*z=2},x=-3..3,y=-3..3,z=-3..3);

> solve({x+y+z=1,x-y+2*z=1,2*x+3*z=0},{x,y,z});
> implicitplot3d({x+y+z=1,x-y+2*z=1,2*x+3*z=0
},x=-3..3,y=-3..3,z=-3..3);

V tomto posledńım př́ıpadě soustava nemá řešeńı, roviny nemaj́ı žádný společný bod.
Soustavy v́ıce rovnic s v́ıce neznámými už si nelze tak jednoduše představit. Řeš́ıme je metodami lineárńı
algebry. Ukažme si možnosti baĺıku linalg při řešeńı soustav.

Př́ıklad:
Následuj́ıćı soustavu lineárńıch algebraických rovnic nejprve zapǐste a vytvořte odpov́ıdaj́ıćı matici a
rozš́ı̌renou matici soustavy (genmatrix, augment). Jakou maj́ı hodnost (př́ıkaz rank)? Rozš́ı̌renou
matici soustavy převed’te na horńı trojúhelńıkový tvar pomoćı ekvivalentńıch úprav, které neměńı
hodnost matice. Připomeňme, že jde o následuj́ıćı úpravy: přehozeńı řádk̊u (swaprow) nebo sloupc̊u
(swapcol), k jednomu řádku (sloupci) přičteńı násobku jiného řádku (sloupce) (př́ıkazy addrow,
addcol), vynásobeńı libovolného řádku nenulovým č́ıslem (mulrow) a vynecháńı nulového řádku
matice (delrows). Nakonec proved’te zpětné dosazeńı (backsub). Kolik má soustava řešeńı? Jak
vypadá nulový prostor matice (=prostor všech řešeńı homogenńı soustavy)? Jaká je jeho dimenze
(nullspace(A, ’nulldim’))?

> restart;

> with(linalg):
> eqns :=
>{3*x1+x2+x3+x4=2,x1-x2+3*x3-x4=-6,2*x1+2*x3=-2,x1+x2-x3+x4=4};

eqns := {3 x1 + x2 + x3 + x4 = 2, x1 − x2 + 3 x3 − x4 = −6, 2 x1 + 2 x3 = −2,

x1 + x2 − x3 + x4 = 4}
> C := genmatrix(eqns, [x1,x2,x3,x4], b);

C :=


3 1 1 1
1 −1 3 −1
2 0 2 0
1 1 −1 1


> evalm(b);

[2, −6, −2, 4]

> rank(C);

2

> Cb:=augment(C,[2,-6,-2,4]);

Cb :=


3 1 1 1 2
1 −1 3 −1 −6
2 0 2 0 −2
1 1 −1 1 4


> rank(Cb);

2

Poznamenejme, že kdybychom jako třet́ı parametr zadali mı́sto vektoru b parametr flag, uložila by
se do matice C celá rozš́ı̌rená matice soustavy.

> C1 := genmatrix(eqns, [x1,x2,x3,x4], flag);

C1 :=


3 1 1 1 2
1 −1 3 −1 −6
2 0 2 0 −2
1 1 −1 1 4


Hodnost matice C je tedy rovna hodnosti rozš́ı̌rené matice soustavy (C|b) a podle Frobeniovy věty
má soustava řešeńı. Protože hodnost matice C je 2 a soustava má 4 neznámé, bude mı́t soustava
nekonečně mnoho řešeńı. Dimenze prostoru všech řešeńı homogenńı soustavy muśı být 2 (později
ověř́ıme v Maplu). Nyńı převedeme matici (C|b) na horńı trojúhelńıkový tvar Cht :

> Cht := swaprow(Cb,1,4);

Cht :=


1 1 −1 1 4
1 −1 3 −1 −6
2 0 2 0 −2
3 1 1 1 2


> Cht := addrow(Cht,1,2,-1);

Cht :=


1 1 −1 1 4
0 −2 4 −2 −10
2 0 2 0 −2
3 1 1 1 2


> Cht := addrow(Cht,1,3,-2);

Cht :=


1 1 −1 1 4
0 −2 4 −2 −10
0 −2 4 −2 −10
3 1 1 1 2


> Cht := addrow(Cht,1,4,-3);

Cht :=


1 1 −1 1 4
0 −2 4 −2 −10
0 −2 4 −2 −10
0 −2 4 −2 −10


> Cht := addrow(Cht,2,3,-1);

Cht :=


1 1 −1 1 4
0 −2 4 −2 −10
0 0 0 0 0
0 −2 4 −2 −10


> Cht := addrow(Cht,2,4,-1);

Cht :=


1 1 −1 1 4
0 −2 4 −2 −10
0 0 0 0 0
0 0 0 0 0


> Cht:=delrows(Cht,3..4);

Cht :=

[
1 1 −1 1 4
0 −2 4 −2 −10

]

> Cht:= mulrow(Cht,2,-1/2);

Cht :=

[
1 1 −1 1 4
0 1 −2 1 5

]
Nejprve jsme přehodili prvńı a čtvrtý řádek, od druhého řádku jsme odečetli prvńı, od třet́ıho řádku
jsme odečetli dvojnásobek prvńıho a od čtvrtého řádku jsme odečetli trojnásobek prvńıho. Vynulu-
jeme 3. a 4. řádek a nulové řádky vynecháme. Druhý řádek vyděĺıme (−2)ma (t.j. násob́ıme −1/2).
Matice Cht je v horńım trojúhelńıkovém tvaru. Zbývá provést zpětný chod:

> x0:= backsub(Cht);

x0 := [−1− t2, 5 + 2 t2 − t1, t2, t1]

> C&*x0-b;

(C &∗ x0 )− b
> evalm(%);

[0, 0, 0, 0]

Zkouška ukázala, že jsme skutečně řešili naši soustavu správně.

Zbývá zodpovědět otázku, jak vypadá nulový prostor matice C, tj. jaký je lineárńı prostor všech
řešeńı př́ıslušné homogenńı soustavy.

> nullspace(C,d);

{[−1, 2, 1, 0], [0, −1, 0, 1]}
> d;

2

Nulový prostor je tedy dvoudimenzionálńı podprostor prostoru R4 a jeho bázi tvoř́ı např́ıklad vektory
(−1, 2, 1, 0)T, (0,−1, 0, 1)T.

Baĺık linalg ale umožňuje hledat řešeńı soustavy hned několika zp̊usoby př́ımo. Řešme soustavu Ux = v :

a) př́ıkaz linsolve

> U:= randmatrix(3,3);

U :=

 −58 −90 53
−1 94 83
−86 23 −84


> v:=randvector(3);

v := [19, −50, 88]

> linsolve(U,v); [
−965021

1645903
,
−8250

71561
,
−788237

1645903

]
> xl:=evalf(%);

xl := [−0.5863170551, −0.1152862593, −0.4789085384]

b) př́ıkaz gausselim provede př́ımý chod Gaussovy eliminace, tedy převede rozš́ı̌renou matici soustavy
na horńı trojúhelńıkový tvar. Chceme-li źıskat řešeńı, muśı vždy následovat zpětný chod Gaussovy
eliminace, tedy př́ıkaz backsub

> Uaug:=augment(U,v);

Uaug :=

 −58 −90 53 19
−1 94 83 −50
−86 23 −84 88


> Uht:=gausselim(Uaug);

Uht :=

 −1 94 83 −50
0 −5542 −4761 2919

0 0
−1645903

5542

788237

5542


> backsub(Uht); [

−965021

1645903
,
−8250

71561
,
−788237

1645903

]
> xg:=evalf(%);

xg := [−0.5863170551, −0.1152862593, −0.4789085384]

c) př́ıkaz gaussjord převede danou matici na diagonálńı s jedničkami na diagonále. Máme-li źıskat
řešeńı soustavy v tomto př́ıpadě, pracujeme s rozš́ı̌renou matićı soustavy. Opět následuje zpětný
chod, ale ten je v tomto př́ıpadě velmi jednoduchý.

> U0:=gaussjord(Uaug);

U0 :=


1 0 0

−965021

1645903

0 1 0
−8250

71561

0 0 1
−788237

1645903


> backsub(U0); [

−965021

1645903
,
−8250

71561
,
−788237

1645903

]
> xj:=evalf(%);

xj := [−0.5863170551, −0.1152862593, −0.4789085384]

> evalm(xl); evalm(xg); evalm(xj);

[−0.5863170551, −0.1152862593, −0.4789085384]

[−0.5863170551, −0.1152862593, −0.4789085384]

[−0.5863170551, −0.1152862593, −0.4789085384]

Uvedenými třemi zp̊usoby jsme skutečně źıskali stejné řešeńı.

> with(plots):

> v:=geneqns(U,[x1,x2,x3],v);

v := {−58 x1 − 90 x2 + 53 x3 = 19, −x1 + 94 x2 + 83 x3 = −50,

−86 x1 + 23 x2 − 84 x3 = 88}

> implicitplot3d(v,x1=-2..0, x2=-1..1,x3=-2..0, axes=boxed);

–2
–1.5

–1
–0.5

0

x1

–1
–0.5

0
0.5

1

x2

–2

–1.5

–1

–0.5

0

Poznamenejme, že matici soustavy U (vektor pravé strany v ) jsme vytvořili př́ıkazem randmatrix
(randvector). Při každém prováděńı tohoto př́ıkazu se tedy vytvoř́ı jiná matice soustavy (jiný vektor
pravé strany). V d̊usledku toho se může stát, že př́ıkaz plot nevykresĺı řešeńı správně. Potom je třeba
upravit intervaly pro zobrazováńı jednotlivých neznámých.

1.4. Determinanty, inverzńı matice, maticové rovnice

> A:=matrix(1,1,[a]);

A :=
[
a
]

> det(A);

a

> G:=matrix(2,2);

G := array(1..2, 1..2, [])

> det(G);

G1, 1 G2, 2 −G1, 2 G2, 1

1.4.1. Determinanty a inverzńı matice Tak bychom mohli pokračovat pro determinanty čtvercových
matic vyšš́ıch řád̊u. Připomeňme, že až do řádu 3 včetně lze determinanty poč́ıtat Sarrusovým pravi-
dlem (pro matici 2 × 2 obecně zapsal toto pravidlo předchoźı př́ıklad). Determinanty matic vyšš́ıch
řád̊u lze poč́ıtat pouze rozvojem podle některého řádku nebo sloupce. Determinant vypočteme pomoćı
př́ıkazu det.

> H:=diag(d1,d2,d3,d4);

H :=


d1 0 0 0
0 d2 0 0
0 0 d3 0
0 0 0 d4


> det(H);

d1 d2 d3 d4

> M:=randmatrix(4,4);

M :=


−85 −55 −37 −35

97 50 79 56
49 63 57 −59
45 −8 −93 92



> det(M);

−5825992

Zmiňme se ještě o př́ıkazu adj. Je-li G daná matice, pak př́ıkaz adj vypoč́ıtá inverzńı matici vynásobenou
determinantem matice G (adj(G) je transponovaná matice algebraických doplňk̊u).

> G0:=adj(G);

G0 :=

[
G2, 2 −G1, 2

−G2, 1 G1, 1

]
> G&*G0;

G&∗G0

> evalm(%); [
G1, 1 G2, 2 −G1, 2 G2, 1 0

0 G1, 1 G2, 2 −G1, 2 G2, 1

]
> (G&*G0)/det(G);

G&∗G0

G1, 1 G2, 2 −G1, 2 G2, 1

> evalm(%); [
1 0
0 1

]
Tedy inverzńı matićı k matici G je matice

1

detG
∗ adj(G).

V Maplu můžeme také poč́ıtat jednotlivé minory (př́ıkaz minor):

> minor(G, 2,2);

[
G1, 1

]
> V:=minor(M, 3,2);

V :=

 −85 −37 −35
97 79 56
45 −93 92


> det(V);

−383352

1.4.2. Inverzńı matice a maticové rovnice Maticová rovnice je rovnice, v ńıž všechny vys-
tupuj́ıćı veličiny včetně neznámé jsou matice. Než takové rovnice budeme řešit, nauč́ıme se, jak
k dané matici pomoćı př́ıkazu inverse najdeme matici inverzńı. Připomeňme, že inverzńı matice
existuje pouze k regulárńı matici, tj. ke čtvercové matici, jej́ıž determinant je nenulový.

> A:=matrix([[0,0,1],[0,1,0],[1,0,0]]);

A :=

 0 0 1
0 1 0
1 0 0


> det(A);

−1

> B:=inverse(A);

B :=

 0 0 1
0 1 0
1 0 0



> evalm(A&*B);  1 0 0
0 1 0
0 0 1


Tedy matice A a B jsou navzájem inverzńı.

Př́ıklad:
Řešme následuj́ıćı maticovou rovnici:

A2 ∗X− 4 ∗B = 3 ∗A ∗X,

kde A, B jsou dané matice. Z této rovnice muśıme nejprve vypoč́ıtat neznámou matici X:

A2 ∗X− 3 ∗A ∗X = 4 ∗B , (A2 − 3 ∗A) ∗X = 4 ∗B , tedy

X = 4 ∗ (A2 − 3 ∗A)−1 ∗B .

> A:=matrix([[1,0,2],[1,0,1],[1,1,0]]);

A :=

 1 0 2
1 0 1
1 1 0


> B:=matrix([[-1,1,1],[1,-1,1],[1,1,-1]]);

B :=

 −1 1 1
1 −1 1
1 1 −1



> X:=4*inverse(A&*A-3*A)&*B;

X :=

4


0
−3

4

−1

4
−1

2

1

2

−1

2
−1

2

1

4

−1

4



 &∗B

> evalm(X);  −4 2 −2
2 −6 2
2 −4 0



2. Př́ıklady

Př́ıklad 1:
Ověřte, že vektory a = (−1; 4; 2;−5; 3)T, b = (1; 1; 0; 0; 4)T, c = (3;−5;−2; 8; 7)T jsou lineárně nezávislé
a vyjádřete vektor v = (1,−4,−2, 5,−3)T jako jejich lineárńı
kombinaci.

Řešeńı:

> restart;

> with(linalg):

> a:=vector([-1,4,2,-5,3]);

a := [−1, 4, 2, −5, 3]

> b:=vector([1,1,0,0,4]);

b := [1, 1, 0, 0, 4]

> c:=vector([3,-5,-2,8,7]);

c := [3, −5, −2, 8, 7]

> v:=vector([1,-4,-2,5,-3]);

v := [1, −4, −2, 5, −3]

> lv:=[a,b,c];

lv := [a, b, c]

> A:=matrix(3,5,lv);

A :=

 −1 4 2 −5 3
1 1 0 0 4
3 −5 −2 8 7


> rank(A);

3

Vektory a, b, c jsou tedy lineárně nezávislé. Na prvńı pohled to poznáme např́ıklad z horńıho
trojúhelńıkového tvaru matice A:

> gausselim(A);  −1 4 2 −5 3
0 5 2 −5 7

0 0
6

5
0

31

5


Chceme-li zjistit, jakou lineárńı kombinaćı vektor̊u a, b, c je vektor v, muśıme naj́ıt koeficienty α, β
a δ lineárńı kombinace α ∗ a + β ∗b + δ ∗ c = v. Znamená to vyřešit nehomogenńı soustavu lineárńıch
algebraických rovnic, přičemž matićı soustavy je AT a pravou stranou je vektor v:

> Av := augment(transpose(A),v);

Av :=


−1 1 3 1

4 1 −5 −4
2 0 −2 −2
−5 0 8 5

3 4 7 −3


> redAv:=gausselim(Av);

redAv :=



−1 1 3 1
0 5 7 0

0 0
6

5
0

0 0 0 0
0 0 0 0


> koef:=backsub(redAv);

koef := [−1, 0, 0]

> alpha:=koef[1];beta:=koef[2];delta:=koef[3];

α := −1

β := 0

δ := 0

Pohledem na zadané vektory se snadno přesvědč́ıme, že vektor v je skutečně roven −a.
Poznamenejme ještě, že všechny vektory, které vzniknou lineárńı kombinaćı vektor̊u a, b, c tvoř́ı lineárńı
podprostor dimenze 3 prostoru R5.

Př́ıklad 2:
Najděte vektor v, který je obrazem vektoru u = (2, 3,−7)T v lineárńım zobrazeńı prostoru R3 do R3

daném matićı

A =

 1 4 2
−3 2 0
−1 3 1

 .

Zjistěte, zda vektor u je jediným vektorem z prostoru R3, který se na vektor v zobraźı. Pokud neńı

jediný, najděte všechny daľśı vektory, které se na tento vektor zobraźı.

Řešeńı:

> restart;

> with(linalg):

> A:=matrix(3,3,[1,4,2,-3,2,0,-1,3,1]);

A :=

 1 4 2
−3 2 0
−1 3 1


> u:=vector([2,3,-7]);

u := [2, 3, −7]

> v:=A&*u;

v := A&∗u
> evalm(v);

[0, 0, 0]

Obrazem vektoru u je tedy nulový vektor v.
Všechny daľśı vektory, které se na vektor v zobraźı, najdeme jako řešeńı homogenńı soustavy (ho-
mogenńı, protože v je nulový vektor) Ax = 0:

> Aht:=gausselim(A);

Aht :=

 1 4 2
0 7 3
0 0 0



> linsolve(A,[0,0,0]); [
t1,

3

2
t1, −

7

2
t1

]
Tedy všechny vektory, které se zobraźı na vektor v, tvoř́ı lineárńı podprostor dimenze 1 prostoru R3.
Jeho báze je tvořena např́ıklad vektorem u (stač́ı polžit t1 = 2).

Př́ıklad 3:
Napǐste matici reprezentuj́ıćı lineárńı zobrazeńı L prostoru R3 do R3 definované vztahem

L(x1, x2, x3)T = (2x1 + x3, x1 − x2 − x3,−x1 + 3x2 + 2x3)T

a najdětete k ńı matici inverzńı.

Řešeńı:

> restart;

> with(linalg):

> A:=matrix(3,3,[2,0,1,1,-1,-1,-1,3,2]);

A :=

 2 0 1
1 −1 −1
−1 3 2


> x:=vector([x1,x2,x3]);

x := [x1 , x2 , x3 ]

> y:=vector([2*x1+x3,x1-x2-x3,-x1+3*x2+2*x3]);

y := [2 x1 + x3 , x1 − x2 − x3 , −x1 + 3 x2 + 2 x3 ]

> evalm(A&*x);

[2 x1 + x3 , x1 − x2 − x3 , −x1 + 3 x2 + 2 x3 ]

Matice A tedy zobraźı vektor x na vektor y. Inverzńı matici sestroj́ıme dvěma zp̊usoby: pomoćı př́ıkazu
gaussjordan a inverse:

> E:=diag(1,1,1);

E :=

 1 0 0
0 1 0
0 0 1


> AE:=augment(A,E);

AE :=

 2 0 1 1 0 0
1 −1 −1 0 1 0
−1 3 2 0 0 1


> gaussjord(AE); 

1 0 0
1

4

3

4

1

4

0 1 0 −1

4

5

4

3

4

0 0 1
1

2
−3

2
−1

2


> Ainv:=inverse(A);

Ainv :=



1

4

3

4

1

4

−1

4

5

4

3

4
1

2
−3

2
−1

2


> evalm(Ainv&*y);

[x1 , x2 , x3 ]

Tedy matice inverzńı k matici A převád́ı vektor y na vektor x.

Př́ıklad 4:
Najděte všechna řešeńı soustavy rovnic

3x1 − 4x2 + 4x3 + 2x4 = 0
7x1 − x2 + 6x3 + 3x4 = 0
−2x1 + x2 − 2x3 − x4 = 0

a určete dimenzi nejmenš́ıho podprostoru prostoru R4, ve kterém všechna řešeńı lež́ı.

Řešeńı:

> restart;

> with(linalg):
> eqns :=
>{3*x1-4*x2+4*x3+2*x4=0,7*x1-x2+6*x3+3*x4=0,-2*x1+x2-2*x3-x4=0};

eqns := {3 x1 − 4 x2 + 4 x3 + 2 x4 = 0, 7 x1 − x2 + 6 x3 + 3 x4 = 0,

−2 x1 + x2 − 2 x3 − x4 = 0}
> C := genmatrix(eqns, [x1,x2,x3,x4]);

C :=

 3 −4 4 2
7 −1 6 3
−2 1 −2 −1


> x:=linsolve(C,[0,0,0]);

x := [−2 t1, t1, t2, 5 t1 − 2 t2]

> rank(C);

2

> n:=4;

n := 4

> dimV_H:=n-rank(C);

dimV H := 2

> basis(C,rowspace);

[[3, −4, 4, 2], [7, −1, 6, 3]]

Řešeńım naš́ı soustavy je vektor x a všechna řešeńı lež́ı v podprostoru R4, jehož báze je tvořena prvńımi
dvěma řádky dané matice. Jeho dimenze je 2.

Př́ıklad 5:

Zjistěte, pro která a ∈ R má soustava lineárńıch rovnic řešeńı:

5x+ 3y + z = a+ 2
x+ 3y + 2z = −2
2x+ 2y + z = a

3x+ y = 2

Pokud řešeńı existuje, najděte ho.

Řešeńı:

> restart;

> with(linalg):

> soustava := {5*x+3*y+z=a+2,x+3*y+2*z=-2,2*x+2*y+z=a,3*x+y=2};
soustava := {5x+ 3 y + z = a+ 2, x+ 3 y + 2 z = −2, 2x+ 2 y + z = a, 3x+ y = 2}

> b:=a->vector([a+2,-2,a,2]);

b := a→ [a+ 2, −2, a, 2]

> A := genmatrix(soustava, [x,y,z]);

A :=


5 3 1
1 3 2
2 2 1
3 1 0


> Ab:=augment(A,b(a));

Ab :=


5 3 1 a+ 2
1 3 2 −2
2 2 1 a
3 1 0 2


Matice (A|b) je rozš́ı̌rená matice soustavy. Ilustrujme si na tomto př́ıkladu, jaký je rozd́ıl v použit́ı
př́ıkaz̊u gausselim a gaussjord.

> Aht:=gausselim(Ab,r);

Aht :=


5 3 1 a+ 2

0
4

5

3

5

3a− 4

5
0 0 0 −2a
0 0 0 0


> r;

3

> rank(A);

2

> backsub(Aht);

Error, (in backsub) inconsistent system

Do matice Aht se uložil horńı trojúhelńıkový tvar rozš́ı̌rené matice soustavy, hodnota parametru r
udává hodnost rozš́ı̌rené matice soustavy. Vzhledem k tomu, že hodnost matice soustavy je 2, vid́ıme,
že soustava může mı́t řešeńı pouze v př́ıpadě, že a je rovno nule. Dř́ıve, než provedeme zpětný chod

pro hodnotu a = 0, vyzkoušejme, jak pracuje př́ıkaz gaussjord. Pro lepš́ı pochopeńı budeme provádět
Gauss-Jordanovu eliminaci rozš́ı̌rené matice soustavy (A|b) postupně po sloupćıch:

> S1:=gaussjord(Ab, 1);

S1 :=



1
3

5

1

5

a+ 2

5

0
12

5

9

5
−12 + a

5

0
4

5

3

5

3a− 4

5

0 −4

5
−3

5

4− 3a

5


> S2:=gaussjord(S1, 2);

S2 :=


1 0 −1

4

−a+ 4

4

0 1
3

4

3a− 4

4

0 0 0 −2a
0 0 0 0


> S3:=gaussjord(S2, 3);

S3 :=


1 0 −1

4

−a+ 4

4

0 1
3

4

3a− 4

4

0 0 0 −2a
0 0 0 0


Provedli jsme redukci prvńıch tř́ı sloupc̊u rozš́ı̌rené matice soustavy, opět vid́ıme, že hodnost matice
A je 2, hodnost (A|b) je tři, neńı-li a = 0. Kdybychom ale zadali př́ıkaz gaussjord bez udáńı sloupce,
v němž má redukce skončit, dostali bychom zcela scestný výsledek, nezávislý na hodnotě parametru
a (Maple totiž v tomto př́ıpadě zcela ignoruje př́ıpadné děleńı nulou a klidně dál redukuje posledńı
sloupec):

> S:=gaussjord(S3);

S :=


1 0 −1

4
0

0 1
3

4
0

0 0 0 1
0 0 0 0


> backsub(S);

Error, (in backsub) inconsistent system

Položme nyńı a = 0 a vypočtěme naše řešeńı pro tuto hodnotu parametru a :

> a:=0;

a := 0

> A0:=augment(A,b(a));

A0 :=


5 3 1 2
1 3 2 −2
2 2 1 0
3 1 0 2


> Aht:=gausselim(A0);

Aht :=


5 3 1 2

0
4

5

3

5
−4

5

0 0 0 0
0 0 0 0


> backsub(Aht,false,t); [

1 +
1

4
t1, −1− 3

4
t1, t1

]
> S0:=gaussjord(A0);

S0 :=


1 0 −1

4
1

0 1
3

4
−1

0 0 0 0
0 0 0 0



> reseni:=backsub(S0,false,t);

reseni :=

[
1 +

1

4
t1, −1− 3

4
t1, t1

]
Druhý parametr false v př́ıkazu backsub znamená, že posledńı sloupec v matici S0 je sloupec pravých
stran, Třet́ı parametr t ř́ıká, pomoćı jakých parametr̊u má být zapsaáno řešeńı (Maple si parametry sám
č́ısluje).
Vid́ıme, že nyńı už oba př́ıkazy gausselim a gaussjord a následný zpětný chod daly stejné řešeńı
závislé na jednom parametru (tj. lineárńı prostor všech řešeńı homogenńı soustavy má dimenzi 1,
jeho bázi tvoř́ı např́ıklad vektor (1,−3, 4)T):

> V:= nullspace(A,p);

V := {[1, −3, 4]}
> p;

1

> basis(V);

{[1, −3, 4]}
Znázorněme si řešeńı pro a = 0 graficky. Hledáme společné body rovin, jejichž rovnice jsou:

5x+ 3y + z = 2 , x+ 3y + 2z = −2 , 2x+ 2y + z = 0 , 3x+ y = 2 .

> with(plots):

> g:=geneqns(A, [x,y,z], b(0));

g := {x+ 3 y + 2 z = −2, 3x+ y = 2, 5x+ 3 y + z = 2, 2x+ 2 y + z = 0}

> implicitplot3d(g,x=-10..10, y=-10..10,z=-10..10, axes=boxed);

–10
–5

0
5

10

x

–10
–5

0
5

10

y

–10

–5

0

5

10

Vid́ıme, že roviny maj́ı společnou pr̊usečnici. Jej́ı parametrické rovnice jsou (viz reseni):

x = 1 +
1

4
t , y = −1− 3

4
t , z = t , kde t je reálný parametr.

Př́ıklad 6:
Vypočtěte determinant matice  f(x) f ′(x) f ′′(x)

g(x) g′(x) g′′(x)
h(x) h′(x) h′′(x)

 ,

kde f(x) = sin2 x, g(x) = cos2 x, h(x) = 1, x ∈ R. Zjistěte, pro která x je determinant roven 1.

Contents First Last Prev Next Back Close Quit



Řešeńı:

> restart;

> with(linalg):

> f:=x->(sin(x))^2; g:=x->(cos(x))^2; h:=x->1;

f := x→ sin(x)2

g := x→ cos(x)2

h := 1
> M:=matrix(3,3,[f(x),diff(f(x),x),diff(f(x),x$2),g(x),diff(g(x),x),dif
> f(g(x),x$2),h(x),diff(h(x),x),diff(h(x),x$2)]);

M :=

 sin(x)2 2 sin(x) cos(x) 2 cos(x)2 − 2 sin(x)2

cos(x)2 −2 sin(x) cos(x) −2 cos(x)2 + 2 sin(x)2

1 0 0


Povšimněte si, jak Matlab zaṕı̌se funkci sin2 x.

> m:=det(M);

m := 0

Determinant je roven 0 pro všechna x ∈ R, matice je tedy singulárńı pro všechna x ∈ R.

Př́ıklad 7:
Najděte matici X tak, aby byla splněna rovnice

X ∗A + A = −A + X , kde A =

(
1 2
3 −1

)
.

Řešeńı:

> restart;

> with(linalg):

> A:=matrix(2,2,[1,2,3,-1]);

A :=

[
1 2
3 −1

]
> X:=matrix(2,2);

X := array(1..2, 1..2, [])

> solve(X*A+A=-A+X,X);

− 2A

A− 1
Vid́ıme, že Maple neumı́ maticovou rovnici zadanou t́ımto zp̊usobem vyřešit. Muśıme nejprve vypoč́ıtat
matici X z rovnice a pak teprve použ́ıt Maple.

X ∗A−X = −2 ∗A => X ∗ (A− E) = −2 ∗A => X = −2 ∗A ∗ (A− E)−1 .

Nejprve vypočteme (A−E)−1, pokud existuje, a pak provedeme násobeńı na pravé straně. Připomeňme,
že E znač́ı jednotkovou matici, tedy diagonálńı matici s jedničkami na diagonále.

> E:=diag(1,1);

E :=

[
1 0
0 1

]
> det(A-E);

−6

Determinant je nenulový, tedy inverzńı matice existuje.

> B:=inverse(A-E);

B :=


1

3

1

3
1

2
0


> X:= 2*A&*B;

X := (2A) &∗B
> X:=evalm(X);

X :=

 8

3

2

3

1 2



3. Cvičeńı
Č́ısla zde uvedených př́ıklad̊u jsou uspořádána tak, že prvńı č́ıslo odkazuje na vzorový vyřešený př́ıklad
v předchoźı části.

Př́ıklad 1.1:
Určete dimenzi nejmenš́ıho podprostoru prostoru R5, ve kterém lež́ı vektory

a = (4; 2; 0;−1; 1)T, b = (−3; 1;−1; 2;−2)T, c = (−2; 4;−2; 3;−3)T,

a napǐste jeden daľśı vektor, který v tomto podprostoru lež́ı. Ověřte!

Př́ıklad 1.2:
Napǐste libovolnou bázi prostoru R4 a vyjádřete vektor a = (2,−2, 1,−3)T pomoćı této báze.

Př́ıklad 1.3:
Zjistěte, zda vektory

a = (−1, 4, 2,−5, 3)T, b = (1, 1, 0, 0, 4)T, c = (3,−5,−2, 8,−1)T

jsou lineárně nezávislé a zjistěte, zda vektor v = (2, 1, 2, 1, 3)T lež́ı v podprostoru, jehož bázi tvoř́ı vektory
a, b, c.

Př́ıklad 1.4:
Z vektor̊u

a = (5,−2, 0, 4)T, b = (3, 1,−2, 5)T, c = (0, 1,−2, 5)T, d = (2,−3, 1, 2)T

vyberte maximálńı počet lineárně nezávislých vektor̊u a zbylé vektory vyjádřete jako lineárńı kombinaci
vybraných vektor̊u. Napǐste jeden libovolný vektor z prostoruR4, který nelež́ı v podprostoru generovaném
vybranými vektory.

Př́ıklad 1.5:
Z vektor̊u

(3, 4, 3, 1)T, (1, 2, 1, 1)T, (2, 5,−4, 3)T, (1, 3, 1, 2)T

vyberte maximálńı počet lineárně nezávislých vektor̊u a doplňte je daľśımi vektory tak, aby tvořily bázi
prostoru R4.

Př́ıklad 2.1:
Najděte vektor v, který je obrazem vektoru u = (1, 1, 1)T v lineárńım zobrazeńı prostoru R3 do R3

reprezentovaném matićı

A :=

 2 −1 0
0 4 −3
0 0 1

 .

Zjistěte, zda vektor u je jediným vektorem z prostoru R3, který se na vektor v zobraźı. Pokud neńı
jediný, najděte všechny daľśı.

Př́ıklad 2.2:
Najděte vektor v, který je obrazem vektoru u = (2, 3, 7)T v lineárńım zobrazeńı prostoru R3 do R4

reprezentovaném matićı

A :=


2 −3 1
3 1 −2
−5 2 0

0 2 −1

 .

Zjistěte, zda vektor u je jediným vektorem z prostoru R3, který se na vektor v zobraźı. Pokud neńı
jediný, najděte všechny daľśı.

Př́ıklad 3.1:
Napǐste matici reprezentuj́ıćı lineárńı zobrazeńı L prostoru R3 do R3 definované vztahem

L(x1, x2, x3)T = (2x1 + x3,−x2 + x3, x1 + 2x2)T

a najdětete k ńı matici inverzńı.

Př́ıklad 3.2:
Dokažte, že zobrazeńı L1 : R3 −→ R4 a zobrazeńı L2 : R4 −→ R3 jsou lineárńı a najděte matici A
reprezentuj́ıćı složené zobrazeńı L1 s L2. Úlohu řešte tak, že

a) naleznete tvar složeného zobrazeńı a pak sestav́ıte jeho matici;

b) setav́ıte nejprve matice jednotlivých zobrazeńı.

L1(x1, x2, x3)T = (x1 − 2x2 + x3, 4x1 + 2x2 − x3,−x1 − 5x2 + 3x3, 2x1 − 3x2 + 8x3)T ;
L2(x1, x2, x3, x4)T = (x1 − 2x2 + x3 − x4, x1 − 7x4, 2x1 − x3 + x4)T.

Př́ıklad 3.3:
Dokažte, že zobrazeńı L1 : R3 −→ R4 a zobrazeńı L2 : R4 −→ R3 jsou lineárńı a najděte matici A
reprezentuj́ıćı složené zobrazeńı L2 s L1. Úlohu řešte tak, že

a) naleznete tvar složeného zobrazeńı a pak sestav́ıte jeho matici;

b) setav́ıte nejprve matice jednotlivých zobrazeńı.

L1(x1, x2, x3)T = (x1 − 2x2 + x3, 4x1 + 2x2 − x3,−x1 − 5x2 + 3x3, 2x1 − 3x2 + 8x3)T ;
L2(x1, x2, x3, x4)T = (x1 − 2x2 + x3 − x4, x1 − 7x4, 2x1 − x3 + x4)T.

Př́ıklad 4.1:
Najděte všechna řešeńı soustavy rovnic

3x+ y + 2z − 5t = −2
6x+ 2y + z − 4t = −4
−3x− y + 2z − 3t = 2

a určete dimenzi a bázi podprostoru prostoru R4, ve kterém lež́ı všechna řešeńı př́ıslušné homogenńı
soustavy.

Př́ıklad 4.2:
Najděte všechna řešeńı soustavy rovnic

5x+ 2y + z + 2t = −1
4x+ y + 2z + 4t = −2
3x+ 2y − z − 2t = 1

a určete dimenzi a bázi podprostoru prostoru R4, ve kterém lež́ı všechna řešeńı př́ıslušné homogenńı
soustavy.

Př́ıklad 4.3:
Najděte všechna řešeńı soustavy lineárńıch rovnic (znázorněte také graficky):

3x− 2y − z = 4
x+ z = 2

x+ y + 3z = 3
4x− y + 2z = 7

a určete dimenzi a některou bázi podprostoru prostoru R3, ve kterém lež́ı všechna řešeńı př́ıslušné ho-
mogenńı soustavy.

Př́ıklad 4.4:
Najděte všechna řešeńı soustavy lineárńıch rovnic (znázorněte také graficky):

3x+ y + z = 3
2x− y + 4z = 7

x+ z = 2
4x+ 2y − 3z = −4

a určete dimenzi a některou bázi podprostoru prostoru R3, ve kterém lež́ı všechna řešeńı př́ıslušné ho-
mogenńı soustavy.

Př́ıklad 4.5:
Najděte všechna řešeńı soustavy rovnic

7x1 + x2 − 3x3 − 6x4 = 0
2x1 + x2 − x3 − 2x4 = 0
−3x1 − 4x2 + 2x3 + 4x4 = 0

a určete dimenzi nejmenš́ıho podprostoru prostoru R4, ve kterém všechna řešeńı lež́ı.

Př́ıklad 5.1:
Zjistěte, pro která a ∈ R má soustava lineárńıch rovnic řešeńı:

3x+ y = 2
5x+ 3y + z = a
2x+ 2y + z = a− 2
−x− 3y − 2z = 2

Všechna řešeńı napǐste a znázorněte graficky.

Př́ıklad 5.2:
Řešte soustavu rovnic

x1 − x2 + x3 − x4 = −1
2x1 + 2x2 + x3 + x4 = 0

x1 − x3 − 2x4 = 0
ax1 + 2x3 + 2x4 = b

s reálnými parametry a, b a proved’te diskusi řešitelnosti vzhledem k těmto parametr̊um. Všechna řešeńı
vypočtěte.

Př́ıklad 5.3:
Řešte soustavu rovnic

2x1 − x2 + x3 = 3
x1 + x2 + 2x3 = 0
x1 − x2 + ax3 = b

s reálnými parametry a, b a proved’te diskusi řešitelnosti vzhledem k těmto parametr̊um. Všechna řešeńı
vypočtěte a graficky znázorněte.

Př́ıklad 5.4:
Zjistěte, pro která reálná a má soustava lineárńıch rovnic řešeńı:

4x+ 3y − 7z − 2u = 5
3x+ 2y − 4z − u = 3
−2x− y + z = a2

8x− 11y + z + 2u = 2

Všechna řešeńı najděte.

Př́ıklad 5.5:
Řešte soustavu rovnic

5x1 + 3x2 + x3 = a+ 1
2x1 + 2x2 + x3 = a− 1
x1 + 3x2 + 2x3 = −2

3x1 + x2 = 2

s reálným parametrem a. Proved’te diskusi řešitelnosti vzhledem k tomuto parametru. Všechna řešeńı
vypočtěte a graficky znázorněte.

Př́ıklad 6.1:
K matici

A =


2 1 0 0
1 0 −1 2
0 0 1 −1
0 1 0 −1



vypočtěte matici inverzńı.

Př́ıklad 6.2:
K matici

A =


2 0 1 0
0 −1 1 0
1 2 0 −1
0 −1 0 1


vypočtěte matici inverzńı.

Př́ıklad 6.3:
Vypočtěte determinant matice  f(x) f ′(x) f ′′(x)

g(x) g′(x) g′′(x)
h(x) h′(x) h′′(x)

 ,

kde f(x) = sin x, g(x) = cos x, h(x) = x, x ∈ R. Zjistěte, pro která x je determinant roven 1.

Př́ıklad 7.1:
Najděte matici X tak, aby byla splněna rovnice

A ∗X + A−1 = B , kde A =

 0 0 2
2 0 0
0 2 0

 , B =

 0 1 0
0 0 1
1 0 0

 .

Př́ıklad 7.2:
Najděte matici X tak, aby byla splněna rovnice

A ∗X + A = X−A , kdekde A =

(
3 4
−1 1

)
.

Př́ıklad 7.3:
Najděte matici X tak, aby byla splněna rovnice

A ∗X−A = E + X , kde (A) =

(
4 −3
−1 −1

)
.

Př́ıklad 7.4:
Najděte matici X tak, aby byla splněna rovnice

A ∗X ∗A−1 = B + E , kde A =

 −1 0 2
2 1 0
0 2 1

 , B =

 3 1 0
0 −1 1
1 2 0

 .

Př́ıklad 7.5:
Najděte matici X tak, aby byla splněna rovnice

A ∗X ∗A−1 = B + X ∗A−1 , kde A =

 −1 0 2
2 1 0
0 2 1

 , B =

 3 1 0
0 −1 1
1 2 0

 .

